

# FreeFem++ Lessons 5-8

F. Hecht and I. Danaila

Laboratoire Jacques-Louis Lions  
Université Pierre et Marie Curie  
Paris, France

<http://www.freefem.org>

<mailto:frederic.hecht@upmc.fr>

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
- 2 Non-linear problems (Lesson 6)
- 3 Mesh adaptation (Lesson 6)
- 4 Incompressible Fluid Dynamics (Lesson 7)
- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
  - Linear elasticity equations
    - Static Linear Lamé equation, weak formulation
    - Representation of the Strain and Stress tensors
    - Solving the static linear elasticity equation in 2d with FreeFem++
    - Solving the static linear elasticity equation in 3d with FreeFem++
    - Solving the time-dependent linear elasticity equation in 2d and 3d

# Linear Lamé equation and Hooke's Law

Let us consider a beam and with transverse section  $\Omega$ , subject to a force  $\mathbf{f}$ , perpendicular to the axis. The components along  $x$  and  $y$  of the displacement  $\mathbf{u}(x)$  in the section  $\Omega$  are governed by the Lamé's system of linear equations.

Remark: we do not use this equation because the associated variational form does not give the correct boundary conditions! We simply use the equilibrium between efforts and constraints:

$$-\nabla \cdot (\boldsymbol{\sigma}) = \rho \mathbf{f} \quad \text{in } \Omega,$$

where the constraint tensor  $\boldsymbol{\sigma}(\mathbf{u})$  is related to deformations using the Hooke's law:

$$\boldsymbol{\sigma}(\mathbf{u}) = \lambda \operatorname{tr}(\boldsymbol{\varepsilon}(\mathbf{u})) \mathbf{I} + 2\mu \boldsymbol{\varepsilon}(\mathbf{u}).$$

$\lambda, \mu$  are the physical Lamé coefficients and the strain tensor is  $\boldsymbol{\varepsilon}(\mathbf{u}) = \frac{1}{2}(\nabla \mathbf{u} + {}^t \nabla \mathbf{u})$ . The corresponding variational (weak) form is:

$$\int_{\Omega} \boldsymbol{\sigma}(\mathbf{u}) : \boldsymbol{\varepsilon}(\mathbf{v}) \, dx - \int_{\Omega} \mathbf{f} \mathbf{v} \, dx - \int_{\partial\Omega} (\boldsymbol{\sigma}(\mathbf{u}) \cdot \mathbf{n}) \mathbf{v} = 0, \quad \mathbf{a} : \mathbf{b} = \sum_{i,j} a_{ij} b_{ij}.$$

Finally, the variational form can be written as :

$$\int_{\Omega} \lambda \nabla \cdot \mathbf{v} \nabla \cdot \mathbf{u} + 2\mu \boldsymbol{\varepsilon}(\mathbf{u}) : \boldsymbol{\varepsilon}(\mathbf{v}) \, dx - \int_{\Omega} \mathbf{f} \mathbf{v} \, dx - \int_{\partial\Omega} (\boldsymbol{\sigma}(\mathbf{u}) \cdot \mathbf{n}) \mathbf{v} = 0.$$

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
  - Linear elasticity equations
  - **Static Linear Lamé equation, weak formulation**
  - Representation of the Strain and Stress tensors
  - Solving the static linear elasticity equation in 2d with FreeFem++
  - Solving the static linear elasticity equation in 3d with FreeFem++
  - Solving the time-dependent linear elasticity equation in 2d and 3d

# Static Linear Lamé equations: weak formulation

Let  $\Omega \subset \mathbb{R}^d$  be a domain with a partition of  $\partial\Omega = \Gamma_d \cup \Gamma_n$ .

Find  $\mathbf{u}$ , the displacement field, such that:

$$-\nabla \cdot \boldsymbol{\sigma}(\mathbf{u}) = \rho \mathbf{f} \text{ in } \Omega, \quad \mathbf{u} = \mathbf{0} \text{ on } \Gamma_d, \quad \boldsymbol{\sigma}(\mathbf{u}) \cdot \mathbf{n} = 0 \text{ on } \Gamma_n, \quad (1)$$

where  $\boldsymbol{\sigma}(\mathbf{u}) = \mathcal{A}\boldsymbol{\varepsilon}(\mathbf{u})$ , with  $\mathcal{A}$  a linear positive operator (symmetric  $d \times d$  matrix) corresponding to material properties. Let us denote  $V_g = \{\mathbf{v} \in H^1(\Omega)^d / \mathbf{v}|_{\Gamma_d} = \mathbf{g}\}$ . The basic (displacement) variational formulation is: find  $\mathbf{u} \in V_0(\Omega)$ , such that:

$$\int_{\Omega} \boldsymbol{\varepsilon}(\mathbf{v}) : \mathcal{A}\boldsymbol{\varepsilon}(\mathbf{u}) = \int_{\Omega} \rho \mathbf{v} \cdot \mathbf{f} + \int_{\Gamma} ((\mathcal{A}\boldsymbol{\varepsilon}(\mathbf{u}))\mathbf{n}) \cdot \mathbf{v}, \quad \forall \mathbf{v} \in V_0(\Omega). \quad (2)$$

The Hooke's law says that  $\mathcal{A} = \lambda \mathbf{I}_d + 2\mu \mathbf{1}_{d,d}$ , where  $\mathbf{I}_d$  is the Identity  $d \times d$  matrix and  $\mathbf{1}_{d,d}$  the constant  $d \times d$  matrix filled with 1.

Question: How to code this equation with FreeFem++?

Remark: the contraction operator  $(:)$  exists, but its priority is low: try to avoid it!

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
  - Linear elasticity equations
  - Static Linear Lamé equation, weak formulation
  - **Representation of the Strain and Stress tensors**
  - Solving the static linear elasticity equation in 2d with FreeFem++
  - Solving the static linear elasticity equation in 3d with FreeFem++
  - Solving the time-dependent linear elasticity equation in 2d and 3d

# Representation of the Strain tensor $\varepsilon$ : the Voigt notation

More details on [https://en.wikipedia.org/wiki/Voigt\\_notation](https://en.wikipedia.org/wiki/Voigt_notation)

We denote by  $\lambda = \lambda$ ,  $\mu = \mu$ ,  $\text{twomul} = 2\lambda + \mu$ , and define

In 2d:

```
func A = [[twomul, lambda, 0. ],  
          [lambda, twomul, 0. ],  
          [ 0. , 0. , mu  ] ] ;
```

```
macro epsV(u1,u2) [dx(u1), dy(u2), dy(u1)+dx(u2)] // EOM
```

```
macro div(u1,u2) ( dx(u1)+dy(u2) ) // EOM
```

In 3d:

```
func A = [[twomul, lambda, lambda, 0. , 0. , 0. ],  
          [lambda, twomul, lambda, 0. , 0. , 0. ],  
          [lambda, lambda, twomul, 0. , 0. , 0. ],  
          [ 0. , 0. , 0. , mu , 0. , 0. ],  
          [ 0. , 0. , 0. , 0. , mu , 0. ],  
          [ 0. , 0. , 0. , 0. , 0. , mu ] ] ;
```

```
macro epsV(u1,u2,u3) [dx(u1), dy(u2), dz(u3),  
                    dz(u2)+dy(u3), dz(u1)+dx(u3), dy(u1)+dx(u2)] // EOM
```

```
macro div(u1,u2,u3) ( dx(u1)+dy(u2)+ dz(u3) ) // EOM
```



- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
  - Linear elasticity equations
  - Static Linear Lamé equation, weak formulation
  - Representation of the Strain and Stress tensors
  - Solving the static linear elasticity equation in 2d with FreeFem++
  - Solving the static linear elasticity equation in 3d with FreeFem++
  - Solving the time-dependent linear elasticity equation in 2d and 3d

# Solving the static linear elasticity equation in 2d with FreeFem++

For the values of material constants, see

<http://www.mstrtech.com/WebPages/matexam.htm>

```
// Steel  $\lambda = 9.695 \cdot 10^{10} \text{ N/m}^2$ ,  
//  $\mu = 7.617 \cdot 10^{10} \text{ N/m}^2$ ,  $\rho = 7700 \text{ kg/m}^3$ .  
real rho = 7700, mu = 7.617e10, lambda = 9.69e10 ;  
real gravity = -9.81, twomul=2*mu+lambda; // Optimisation  
cout << "lambda,mu,gravity_=" << lambda << " " << mu << " " << gravity << endl;
```

The FreeFem++ code:

```
int[int] labs=[1,1,1,2];  
mesh Th=square(50,5,[x*10,y],label=labs);  
fespace Vh(Th,[P1,P1]);  
  
Vh [u1,u2], [v1,v2],[un1,un2],[up1,up2];  
solve Lamé([u1,u2],[v1,v2])= int2d(Th) ( epsV(u1,u2)'*A*epsV(v1,v2))  
- int2d(Th) ( rho*gravity*v2) + on(2,u1=0,u2=0) ;  
  
real dmax= u1[].linfty, coef= 3/dmax;  
cout << "max_displacement_=" << dmax << "coef_" << coef << endl;  
mesh Thm = change(movemesh(Th,[x+u1*coef,y+u2*coef]),fregion=1);  
plot(Th,Thm,wait=1,cmm="coef_amplification_="+coef);
```

Run:Beam-Static-2d.edp

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
  - Linear elasticity equations
  - Static Linear Lamé equation, weak formulation
  - Representation of the Strain and Stress tensors
  - Solving the static linear elasticity equation in 2d with FreeFem++
  - Solving the static linear elasticity equation in 3d with FreeFem++
  - Solving the time-dependent linear elasticity equation in 2d and 3d

# Solving the static linear elasticity equation in 3d with FreeFem++

Using the same physical parameters:  $\lambda = 9.695 \cdot 10^{10} \text{ N/m}^2$ ,  $\mu = 7.617 \cdot 10^{10} \text{ N/m}^2$ ,  
 $\rho = 7700 \text{ kg/m}^3$ .

The FreeFem++ code:

```
int[int] labs=[1,1,1,2,1,1];
mesh3 Th=cube(50,5,5,[x*10,y,z],label=labs);

fespace Vh(Th,[P1,P1,P1]);
Vh [u1,u2,u3],[v1,v2,v3],[un1,un2,un3],[up1,up2,up3];
solve Lamé([u1,u2,u3],[v1,v2,v3])=
  int3d(Th)( epsV(u1,u2,u3)'*A*epsV(v1,v2,v3))
  - int3d(Th)( rho*gravity*v3)
  + on(2,u1=0,u2=0,u3=0)
  ;
real dmax= u1[].lminfty, coef= 5/dmax;
cout << "max displacement=" << dmax << "coef=" << coef << endl;
int[int] llm=[1,3]; // just to change the color of plot mesh
mesh3 Thm=movemesh(Th,[x+u1*coef,y+u2*coef,z+u3*coef],label=llm);
plot(Th,Thm, wait=1,cmm="coef amplification="+coef);
```

Run:Beam-Static-3d.edp

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
  - Linear elasticity equations
  - Static Linear Lamé equation, weak formulation
  - Representation of the Strain and Stress tensors
  - Solving the static linear elasticity equation in 2d with FreeFem++
  - Solving the static linear elasticity equation in 3d with FreeFem++
  - Solving the time-dependent linear elasticity equation in 2d and 3d

# Solving the time-dependent linear elasticity equation in 2d and 3d

The problem is (strong formulation)

$$\rho \partial_{tt} \mathbf{u} - \nabla \cdot (\boldsymbol{\sigma}(\mathbf{u})) = \rho \mathbf{f} \quad \text{in } \Omega.$$

We use a classical explicit 2nd order finite difference scheme for the time derivative:

$$\rho \frac{u^{n+1} - 2u^n + u^{n-1}}{(\delta t)^2} - \nabla \cdot (\boldsymbol{\sigma}(\mathbf{u}^n)) = \rho \mathbf{f}^n$$

Let us denote by  $u_n = u^{n+1}$ ,  $u = u^n$ ,  $u_p = u^{n-1}$ ;  
the matrix formulation of the problem is:

$$u_n = M^{-1}b, \quad b = M(-u_p) + Au + r, \quad +B.C$$

$$M \equiv \int_{\Omega} \rho \frac{\mathbf{u} \cdot \mathbf{v}}{(\delta t)^2} + B.C, \quad A \equiv \int_{\Omega} -\varepsilon(\mathbf{v}) : \mathcal{A} \varepsilon(\mathbf{u}) + 2\rho \frac{\mathbf{u} \cdot \mathbf{v}}{(\delta t)^2}, \quad r \equiv \int_{\Omega} \rho g \mathbf{e}_3 \cdot \mathbf{v} + B.C$$

# Program for the time-dependent linear elasticity equation in 2d and 3d

```
include "Beam-Static-2d.edp"  
real dt =1e-5,      rhodt2= rho/dt/dt;  
varf vA( [u1,u2],[v1,v2]) = int2d(Th) ( -1*epsV(u1,u2)'*A*epsV(v1,v2)  
      + 2*rhodt2*[u1,u2]'*[v1,v2]);  
varf vM( [u1,u2],[v1,v2]) = int2d(Th) ( rhodt2*[u1,u2]'*[v1,v2])  
      + on(2,u1=0,u2=0);  
varf vB( [u1,u2],[v1,v2]) = int2d(Th) ( rho*[0,gravity]'*[v1,v2])  
      + on(2,u1=0,u2=0);  
  
matrix AA=vA(Vh,Vh),      M=vM(Vh,Vh, solver=CG);  
real[int] Rhs = vB(0,Vh);  
  
func BB=[[-0.5,-7],[10.5,1.4]]; // for fixe bounding box of the plot ..  
up1[]=u1[]=0;  
for(int i=0; i<100000; ++i) {  
  real[int] b = AA*u1[];  up1[]=-up1[];  b += Rhs;  b += M*up1[];  
  un1[]= M^-1*b;  
  up1[]=u1[];  u1[]=un1[];  
  if(i%100==0) { cout << i << "\n" << u1[]<< endl;  
    mesh Thmm =movemesh(Th,[x+u1*coef,y+u2*coef]);  
  }  
}
```

Run:Beam-Vibration2d.edp

Run:Beam-Vibration3d.edp

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
- 2 Non-linear problems (Lesson 6)
- 3 Mesh adaptation (Lesson 6)
- 4 Incompressible Fluid Dynamics (Lesson 7)
- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)



## 2 Non-linear problems (Lesson 6)

- Algorithms for solving non-linear problems: fixed point algorithm, Newton method
- Example of a non-linear problem: the Minimal Surface problem
- A fixed-point method to solve the Minimal Surface problem
- A Newton method to solve the Minimal Surface problem

# The fixed-point algorithm

Consider the non-linear problem  $F(u, u) = 0$ , with  $F(\cdot, u)$  affine with respect to the first variable. To find a solution, you can try the following basic method, with no guaranty of convergence:

- 1 set  $u^0$  an initial guess
- 2 do (iterations following  $n$ )
  - 1 find  $u^{n+1}$ , the solution to  $F(u^{n+1}, u^n) = 0$ ,
  - 2 if(  $\|u^{n+1} - u^n\| < \varepsilon$ ) break;

The difficulty in this algorithm is to find an initial guess; sometimes this algorithm explodes. The convergence is generally slow.

# The Newton method

To solve  $F(u) = 0$  we can also use the Newton method ( $DF$  is the differential of  $F$ ):

- 1 set  $u^0$ , an initial guess
- 2 do (iterations following  $n$ )
  - 1 find  $w^n$ , solution to  $DF(u^n)w^n = F(u^n)$
  - 2  $u^{n+1} = u^n - w^n$
  - 3 if(  $\|w^n\| < \varepsilon$ ) break;

The Optimized Newton Method:

if  $F = C + L + N$ , with  $C$  the constant,  $L$  the linear, and  $N$  the non-linear part of  $F$ .

We obtain that  $DF = L + DN$  and the Newton method can be written as:

$$DF(u^n)u^{n+1} = DF(u^n)u^n - F(u^n) = DN(u^n)u^n - N(u^n) - C.$$

The new version of the algorithm is:

- 1 do
  - 1 find  $u^{n+1}$  solution to
$$DF(u^n)u^{n+1} = DN(u^n)u^n - N(u^n) - C = DF(u^n)u^n - F(u^n)$$
  - 2 if(  $\|u^{n+1} - u^n\| < \varepsilon$ ) break;

The weakness of this algorithm is the need to start from an initial guess sufficiently close to a solution.

## 2 Non-linear problems (Lesson 6)

- Algorithms for solving non-linear problems: fixed point algorithm, Newton method
- Example of a non-linear problem: the Minimal Surface problem
- A fixed-point method to solve the Minimal Surface problem
- A Newton method to solve the Minimal Surface problem

# Example of a non-linear problem: the Minimal Surface problem

Let us solve the following geometrical problem: Find a function  $u : \Omega \mapsto \mathbb{R}$ , where  $u$  is given on  $\Gamma = \partial\Omega$ , (i.e.  $u|_{\Gamma} = g$ ) such as the area of the surface  $S$ , parametrized by  $(x, y) \in \Omega \mapsto (x, y, u(x, y))$  is minimal.

The mathematical formulation of the problem is:

$$\arg \min J(u) = \int_{\Omega} \left\| \begin{pmatrix} 1 \\ 0 \\ \partial_x u \end{pmatrix} \times \begin{pmatrix} 0 \\ 1 \\ \partial_y u \end{pmatrix} \right\| d\Omega = \int_{\Omega} \sqrt{1 + (\partial_x u)^2 + (\partial_y u)^2} d\Omega.$$

The Euler-Lagrange equation associated to the minimization of  $J(u)$  is:

$$\forall v/v|_{\Gamma} = 0 \quad : \quad DJ(u)v = \int_{\Omega} \frac{(\partial_x v \partial_x u + \partial_y v \partial_y u)}{\sqrt{1 + (\partial_x u)^2 + (\partial_y u)^2}} d\Omega = 0.$$

We consider the case:  $\Omega = ]0, \pi[^2$  and  $g(x, y) = \cos(nx) \cos(ny)$ ,  
 $n = 1$  (simplest problem) and  $n = 2$  or  $4$  (harder to solve).

We shall use the fixed-point algorithm and the Newton method.

## 2 Non-linear problems (Lesson 6)

- Algorithms for solving non-linear problems: fixed point algorithm, Newton method
- Example of a non-linear problem: the Minimal Surface problem
- A fixed-point method to solve the Minimal Surface problem
- A Newton method to solve the Minimal Surface problem

# A fixed-point method to solve the Minimal Surface problem

```
int nn=100,n=4; // n= 1 ,2,4
int[int] l1=[1,1,1,1];
mesh Th= square(nn,nn,[x*pi,y*pi],label=l1);
func g = cos(n*x)*cos(n*y);
fespace Vh(Th,P1);
Vh un,u,v;
for(int i=0; i< 1000; ++i)
{
  verbosity =0;
  solve Pb(un,v) = int2d(Th) ( (dx(un)*dx(v)+ dy(un)*dy(v))
                               / sqrt( 1. + (dx(u)*dx(u)+ dy(u)*dy(u)) ) )
  + on(1,un = g);
  real J = int2d(Th) ( sqrt( 1. + (dx(un)*dx(un)+ dy(un)*dy(un)) ) );
  plot(un,dim=3,fill=1, wait=0);
  u[]-=un[]; // diff
  real err= u[].linfy;
  cout << "iter" << i << " " << err << " " << "J" << J << endl;
  if( err < 1e-6) break;
  u[]=un[]; }
}
```

Run:Min-Surf-FixPoint.edp

## 2 Non-linear problems (Lesson 6)

- Algorithms for solving non-linear problems: fixed point algorithm, Newton method
- Example of a non-linear problem: the Minimal Surface problem
- A fixed-point method to solve the Minimal Surface problem
- A Newton method to solve the Minimal Surface problem



# A Newton method to solve the Minimal Surface problem

```
// macro of compute all differentiel
macro grad2(u,v) ( dx(u)*dx(v)+ dy(u)*dy(v) ) //
macro JJ(u) sqrt( 1. + grad2(u,u) ) //
macro dJJ(u,du) ( grad2(u,du) / JJ(u) ) //
macro ddJJ(u,du,ddu) ( grad2(ddu,du)/JJ(u)
                      - (grad2(u,du)*grad2(u,du)/JJ(u)^3) ) // For Newton
fespace Vh(Th,P1);
Vh u,v,w;
// Stating point ...
solve Pb0(u,v) = int2d(Th) ( grad2(u,v) ) + on(1,u = g);
plot(u,dim=3,wait=0);
// Newton loop
for(int i=0; i< 100; ++i)
{ verbosity =0;
  solve PbTangent(w,v) = int2d(Th) ( ddJJ(u,w,v) ) - int2d(Th) ( dJJ(u,v) )
  + on(1,2,3,4,w = 0);
  u[] -=w[];
  real J = int2d(Th) ( JJ(u) );
  plot(u,dim=3,fill=1, wait=0,cmm="_J_"+"J");
  real err= w[].linfty;
  cout << "_iter_" << i << "_err_" << err <<"_ " << "_J_" << J << endl;
  if( err < 1e-6 || err >100) break; }
```

Run:Min-Surf-Newton.edp

Run:Min-Surf-Newton-V2.edp

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
- 2 Non-linear problems (Lesson 6)
- 3 Mesh adaptation (Lesson 6)**
- 4 Incompressible Fluid Dynamics (Lesson 7)
- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)

## 3 Mesh adaptation (Lesson 6)

- Metrics and Unit Mesh
- Metrics and norms
- Solving the 2d Poisson equation using mesh adaptation
- Solving the 3d Poisson equation using mesh adaptation
- A Newton method with mesh adaptation for the Minimal Surface problem

# Mesh adaptivity: Metrics and Unit Mesh

In Euclidean geometry the length  $|\gamma|$  of a curve  $\gamma$  of  $\mathbb{R}^d$  parametrized by  $\gamma(t)_{t=0..1}$  is

$$|\gamma| = \int_0^1 \sqrt{\langle \gamma'(t), \gamma'(t) \rangle} dt$$

We introduce the metric  $\mathcal{M}(x)$  as a field of  $d \times d$  symmetric positive definite matrices, and the length  $\ell$  of  $\Gamma$  w.r.t  $\mathcal{M}$  is:

$$\ell = \int_0^1 \sqrt{\langle \gamma'(t), \mathcal{M}(\gamma(t))\gamma'(t) \rangle} dt$$

The key-idea is to construct a mesh for which the lengths of the edges are close to 1, accordingly to  $\mathcal{M}$ .

# Metrics intersection

For a metric  $\mathcal{M}$ , the unit ball  $\mathcal{B}\mathcal{M}$  (obtained by plotting the maximum mesh size in all directions), is an ellipse.

If you have two unknowns  $u$  and  $v$ , we just compute the metrics  $\mathcal{M}_u$  and  $\mathcal{M}_v$ , find a metric  $\mathcal{M}_{uv}$ , called intersection, defined by the biggest ellipse such that:

$$\mathcal{B}(\mathcal{M}_{uv}) \subset \mathcal{B}(\mathcal{M}_u) \cap \mathcal{B}(\mathcal{M}_v)$$

## Example of an adaptive mesh

$$u = (10x^3 + y^3) + \tanh(500(\sin(5y) - 2x));$$

$$v = (10y^3 + x^3) + \tanh(5000(\sin(5y) - 2x));$$

Run:Adapt-uv.edp

## 3 Mesh adaptation (Lesson 6)

- Metrics and Unit Mesh
- **Metrics and norms**
- Solving the 2d Poisson equation using mesh adaptation
- Solving the 3d Poisson equation using mesh adaptation
- A Newton method with mesh adaptation for the Minimal Surface problem

# Building the metrics from the solution $u$

For  $P_1$  continuous Lagrange finite elements, the optimal metric norms for the interpolation error (used in the function `adaptmesh` in FreeFem++) are:

- $L^\infty$  :  $\mathcal{M} = \frac{1}{\varepsilon} |\nabla \nabla u| = \frac{1}{\varepsilon} |\mathcal{H}|$ , where  $\mathcal{H} = \nabla \nabla u$
- $L^p$  :  $\mathcal{M} = \frac{1}{\varepsilon} |\det(\mathcal{H})|^{\frac{1}{2p+2}} |\mathcal{H}|$ , (result by F. Alauzet, A. Dervieux)

For the norm  $W^{1,p}$ , the optimal metric  $\mathcal{M}_\ell$  for the  $P_\ell$  Lagrange finite element is given by (with only acute triangles) (thanks to J-M. Mirebeau)

$$\mathcal{M}_{\ell,p} = \frac{1}{\varepsilon} (\det \mathcal{M}_\ell)^{\frac{1}{\ell p + 2}} \mathcal{M}_\ell$$

and (see `MetricPk` plugin and function )

- for  $P_1$ :  $\mathcal{M}_1 = \mathcal{H}^2$  (sub-optimal: for acute triangles, take  $\mathcal{H}$ )

- for  $P_2$ :  $\mathcal{M}_2 = 3 \sqrt{\begin{pmatrix} a & b \\ b & c \end{pmatrix}^2 + \begin{pmatrix} b & c \\ c & a \end{pmatrix}^2}$  with

$$D^{(3)}u(x, y) = (ax^3 + 3bx^2y + 3cxy^2 + dy^3)/3!,$$

Run: `adapt.edp`

Run: `AdaptP3.edp`



## 3 Mesh adaptation (Lesson 6)

- Metrics and Unit Mesh
- Metrics and norms
- Solving the 2d Poisson equation using mesh adaptation
- Solving the 3d Poisson equation using mesh adaptation
- A Newton method with mesh adaptation for the Minimal Surface problem

# The problem with a corner singularity (adaptivity with metrics)

The domain is a L-shaped polygon  $\Omega = ]0, 1[ \setminus ]\frac{1}{2}, 1]^2$  and the PDE is

$$\text{find } u \in H_0^1(\Omega) \text{ such that } -\Delta u = 1 \text{ in } \Omega.$$

The solution has a singularity at the re-entrant angle and we wish to capture it numerically.

# The problem with a corner singularity (FreeFem++ program)

```
int[int] lab=[1,1,1,1];
mesh Th = square(6,6, label=lab);
Th=trunc(Th,x<0.5 | y<0.5, label=1);
fespace Vh(Th,P1);
Vh u,v;
real error=0.01;
problem Problem1(u,v, solver=CG, eps=1.0e-6) =
    int2d(Th) ( dx(u)*dx(v) + dy(u)*dy(v) ) - int2d(Th) ( v)
    + on(1,u=0);
for (int i=0;i< 7;i++)
{
    Problem1; // solving the pde problem
    plot(u,Th,wait=1);

    Th=adaptmesh(Th,u,err=error,nbvx=100000); // the adaptation with Hessian of u
    u=u;
} ;
```

Run:CornerLap.edp

## 3 Mesh adaptation (Lesson 6)

- Metrics and Unit Mesh
- Metrics and norms
- Solving the 2d Poisson equation using mesh adaptation
- Solving the 3d Poisson equation using mesh adaptation
- A Newton method with mesh adaptation for the Minimal Surface problem

# Solving the 3d Poisson equation using mesh adaptation

```
load "msh3" load "tetgen" load "mshmet" load "medit"  
int nn = 6; int[int] l1=[1,1,1,1,1,1];  
mesh3 Th3=trunc( cube(nn,nn,nn, label=l1)  
                , (x<0.5) | (y < 0.5) | (z < 0.5), label=1);  
fespace Vh(Th3,P1); Vh u,v,h;  
macro Grad(u) [dx(u),dy(u),dz(u)] // EOM  
problem Poisson(u,v, solver=CG) = int3d(Th3) ( Grad(u)'*Grad(v) )  
    -int3d(Th3) ( 1*v ) + on(1,u=0);  
real errm=1e-2; // level of error  
for(int ii=0; ii<5; ii++)  
{ Poisson;  
  cout << "u_min, u_max=" << u[].min << " " << u[].max << endl;  
  h=0. ; // for resizing h  
  h[]=mshmet(Th3,u,normalization=1,aniso=0,nbregul=1,hmin=1e-3,hmax=0.3,err=errm);  
  cout << "h_min, h_max=" << h[].min << " " << h[].max << " " << h[].n << " "  
    << Th3.nv << " " << Th3.nt << endl;  
  plot(u,wait=1);  
  errm*= 0.8; // change the level of error  
  Th3=tetgreconstruction(Th3,switch="raAQ",sizeofvolume=h*h*h/6.); }  
Poisson;  
medit ("U-adap-iso-"+5,Th3,u,wait=1);
```

Run:Laplace-Adapt-3d.edp

## 3 Mesh adaptation (Lesson 6)

- Metrics and Unit Mesh
- Metrics and norms
- Solving the 2d Poisson equation using mesh adaptation
- Solving the 3d Poisson equation using mesh adaptation
- A Newton method with mesh adaptation for the Minimal Surface problem

# A Newton method with mesh adaptation for the Minimal Surface problem

```
real errA=0.1;
for(int adap=0; adap<7; adap++)
{ verbosity =0;
  for(int i=0; i< 100; ++i)
  { // ALGO NEWTON OPTIMIZE
    solve PbTangent (un,v) = int2d(Th) ( ddJJ(u,un,v) ) - int2d(Th) ( ddJJ(u,u,v) -
      dJJ(u,v) )
    + on(1,2,3,4,un = g);
    w[] =u[] -un[]; u[]=un[];
    real J = int2d(Th) ( JJ(u) );
    plot(u,dim=3,fill=1, wait=0,cmm="_J_"+"J");
    real err= w[].linfty;
    cout << "_iter_" << i << "_" << err << "_" << "_J_" << J << "_" << "_" << errA
      << endl;
    if( err < 1e-5) break;
    assert(err<10); }
  cout << "adaptmesh_" << endl;
  Th = adaptmesh(Th,u,err=errA,nbvx=100000,ratio = 1.5);
  plot(Th,WindowIndex=1);
  v=0;u=u; w=0; un=un; // resize
  errA = errA/2;
}
```

Run:Min-Surf-Newton-Adapt.edp

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
- 2 Non-linear problems (Lesson 6)
- 3 Mesh adaptation (Lesson 6)
- 4 Incompressible Fluid Dynamics (Lesson 7)**
- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)



- 4 Incompressible Fluid Dynamics (Lesson 7)
  - The stress tensor for a Newtonian fluid
  - Stokes equation: variational formulation
  - Incompressible Navier-Stokes equation: steady states

# The stress tensor for a Newtonian fluid

In the domain  $\Omega$  of  $\mathbb{R}^d$ , we denote by  $\mathbf{u}$  the velocity field,  $p$  the pressure field and  $\mu$  the dynamic viscosity. The classical mechanical stress  $\boldsymbol{\sigma}^*$  of the fluid is:

$$\boldsymbol{\sigma}^*(\mathbf{u}, p) = 2\mu\mathbb{D}(\mathbf{u}) - pI_d, \quad \mathbb{D}(\mathbf{u}) = \frac{1}{2}(\nabla\mathbf{u} + {}^t\nabla\mathbf{u}) \quad (3)$$

or in the math formulation:

$$\boldsymbol{\sigma}^\bullet(\mathbf{u}, p) = \mu\nabla\mathbf{u} - pI_d \quad (4)$$

So  $\boldsymbol{\sigma}$  is one of these two stress tensors. Remark: if  $\nabla \cdot \mathbf{u} = 0$  and  $\mu$  is constant, then  $\nabla \cdot 2\mathbb{D}(\mathbf{u}) = \mu\nabla \cdot \nabla\mathbf{u} + \mu\nabla \cdot {}^t\nabla\mathbf{u} = \mu\nabla \cdot \nabla\mathbf{u} + \mu\underbrace{\nabla \cdot \nabla\mathbf{u}}_{=0} = \mu\nabla^2\mathbf{u} = \mu\Delta\mathbf{u}$ .

**Stokes Equation:** find the velocity field  $\mathbf{u}$  and the pressure field  $p$ , satisfying :

$$-\nabla \cdot \boldsymbol{\sigma}(\mathbf{u}, p) = \mathbf{f} \quad (5) \quad \text{or} \quad -\mu\Delta\mathbf{u} + \nabla p = \mathbf{f} \quad (7)$$

$$-\nabla \cdot \mathbf{u} = 0 \quad (6) \quad -\nabla \cdot \mathbf{u} = 0 \quad (8)$$

where  $\mathbf{f}$  is the density of external forces.

+ Boundary conditions that will be defined through the variational (weak) form.

- 4 Incompressible Fluid Dynamics (Lesson 7)
  - The stress tensor for a Newtonian fluid
  - Stokes equation: variational formulation
  - Incompressible Navier-Stokes equation: steady states

# Stokes equation: variational formulation

Mechanical variational form of the Stokes equation:

$$\forall \mathbf{v}, q; \quad \int_{\Omega} 2\mu \mathbb{D}(\mathbf{u}) : \mathbb{D}(\mathbf{v}) - q \nabla \cdot \mathbf{u} - p \nabla \cdot \mathbf{v} = \int_{\Omega} \mathbf{f} \cdot \mathbf{v} + \int_{\Gamma} {}^t \mathbf{n} \boldsymbol{\sigma}^*(\mathbf{u}, p) \mathbf{v}$$

Mathematical variational form of the Stokes equation:

$$\forall \mathbf{v}, q; \quad \int_{\Omega} \mu \nabla \mathbf{u} : \nabla \mathbf{v} - q \nabla \cdot \mathbf{u} - p \nabla \cdot \mathbf{v} = \int_{\Omega} \mathbf{f} \cdot \mathbf{v} + \int_{\Gamma} {}^t \mathbf{n} \boldsymbol{\sigma}^\bullet(\mathbf{u}, p) \mathbf{v}$$

But remember that  ${}^t \mathbf{n} \boldsymbol{\sigma}^\bullet(\mathbf{u}, p)$  are boundary density forces  $\mathbf{f}_\Gamma$  and not  ${}^t \mathbf{n} \boldsymbol{\sigma}^*(\mathbf{u}, p)$ .

If the B.C. is  $\mathbf{u} = \mathbf{u}_\Gamma$  for all boundaries, then the two formulations are identical.

The pressure  $p$  is defined up to an additive constant and the weak formulation can use a small regularization (to remove the problem of the additive constant and impose a zero mean value for the pressure).

$$\forall \mathbf{v} \in (H_0^1)^d, q \in L^2; \quad \int_{\Omega} \mu \nabla \mathbf{u} : \nabla \mathbf{v} - q \nabla \cdot \mathbf{u} - p \nabla \cdot \mathbf{v} - \varepsilon p q = \int_{\Omega} \mathbf{f} \cdot \mathbf{v}$$

# Solving the Stokes equation with FreeFem++: the entrained cavity flow

```
int nn=10;
mesh Th=square(nn,nn);
macro grad(u) [dx(u),dy(u)] //
macro Grad(u1,u2) [grad(u1),grad(u2)] //
macro D(u1,u2) [ [dx(u1),(dy(u1)+dx(u2))*0.5] , [(dy(u1)+dx(u2))*0.5,dy(u2)] ] //
macro div(u1,u2) (dx(u1)+dy(u2))//
real epsp =1e-8, mu = 1;
```

Choose the correct finite-element couple for velocity and pressure: (P2,P1), (P1b,P1), (P1nc, P0), ...

```
fespace Vh(Th,P2); fespace Ph(Th,P1); // Taylor Hood Finite element
```

```
Vh u1,u2, v1,v2; Ph p,q ;
solve Stokes([u1,u2,p],[v1,v2,q]) =
int2d(Th) ( mu*(Grad(u1,u2):Grad(v1,v2))
           - div(u1,u2)*q - div(v1,v2)*p -epsp*p*q )
+ on(1,2,4,u1=0,u2=0) + on(3,u1=1,u2=0) ;
plot([u1,u2],p,wait=1);
cout << " _mean_value_pressure=_ " << int2d(Th) (p)/Th.area<<endl;
```

Run:Stokes-Cavity.edp

- 4 Incompressible Fluid Dynamics (Lesson 7)
  - The stress tensor for a Newtonian fluid
  - Stokes equation: variational formulation
  - Incompressible Navier-Stokes equation: steady states

# Incompressible Navier-Stokes equation: steady states

Computing steady states of the Incompressible Navier-Stokes equation: In the domain  $\Omega$  of  $\mathbb{R}^d$ , find the velocity field  $\mathbf{u}$  and the pressure field  $p$ , solution to:

$$(\mathbf{u} \cdot \nabla) \mathbf{u} - \nabla \cdot \boldsymbol{\sigma}(\mathbf{u}, p) = \mathbf{f}, \quad (9)$$

$$-\nabla \cdot \mathbf{u} = 0, \quad (10)$$

+ Boundary conditions.

First idea: use the Optimized Newton Method (see page 19)! the only non-linear term is  $N(\mathbf{u}) = (\mathbf{u} \cdot \nabla) \mathbf{u}$  and the differential is  $DN(\mathbf{u})\mathbf{w} = (\mathbf{u} \cdot \nabla) \mathbf{w} + (\mathbf{w} \cdot \nabla) \mathbf{u}$ ; so, the iteration  $\ell$  of the Newton algorithm is:

Find  $\mathbf{u}^\ell, p^\ell$  such that

$$\begin{aligned} \forall \mathbf{v} \in (H_0^1)^d, \quad q \in L^2; \\ \int_{\Omega} \mu(\nabla \mathbf{u}^\ell : \nabla \mathbf{v}) - q \nabla \cdot \mathbf{u}^\ell - p^\ell \nabla \cdot \mathbf{v} + \mathbf{v} \cdot ((\mathbf{u}^\ell \cdot \nabla) \mathbf{u}^{\ell-1} + (\mathbf{u}^{\ell-1} \cdot \nabla) \mathbf{u}^\ell) - \varepsilon p^\ell q \\ = \int_{\Omega} \mathbf{v} \cdot ((\mathbf{u}^{\ell-1} \cdot \nabla) \mathbf{u}^{\ell-1}) + \mathbf{f} \cdot \mathbf{v} \end{aligned}$$

# Incompressible Navier-Stokes equation: steady states of the entrained cavity

```
real epsp =1e-8, mu = 1./Reynold , eps= 1e-5;
Vh u1=0,u2=0, un1,un2, v1,v2; Ph p,pn,q ;
macro UGradW( u1,u2, w1,w2) [ [u1,u2]'*grad(w1) , [u1,u2]'*grad(w2)]//
verbosity=0;
for(int iter=0; iter<20; ++iter)
{ // DF(u)un = DN(u)u - N(u) = UGradW(u1,u2,u1,u2)
  solve Tangent ([un1,un2,pn], [v1,v2,q]) =
    int2d(Th) ( UGradW(u1,u2, un1,un2)'*[v1,v2]
      + UGradW(un1,un2, u1,u2)'*[v1,v2]
      + mu*(Grad(un1,un2):Grad(v1,v2))
      - div(un1,un2)*q - div(v1,v2)*pn -epsp*pn*q
    )
  - int2d(Th) ( UGradW(u1,u2,u1,u2)'*[v1,v2] )
+ on(1,2,4,un1=0,un2=0) + on(3,un1=1,un2=0) ;
u1[]-=un1[]; u2[]-=un2[]; p[]-=pn[]; //diff err
real err1=u1[].linfty, err2 =u2[].linfty , errp = p[].linfty;
cout << "iter=" <<iter << "errs=" << err1 << " " << err2 << " " << errp <<
  endl;
u1[]=un1[]; u2[]=un2[]; p[]=pn[];
plot ([u1,u2],p,wait=1,cmm=iter);
if( err1 < eps & err2 < eps & errp < eps) break;
}
```

Run:Navier-Stokes-Cavity.edp



# Incompressible Navier-Stokes equation: unsteady computations

In the domain  $\Omega$  of  $\mathbb{R}^d$ , find the velocity field  $\mathbf{u}$  and the pressure field  $p$ , solution to:

$$\partial_t \mathbf{u} + (\mathbf{u} \cdot \nabla) \mathbf{u} - \nabla \cdot \boldsymbol{\sigma}(\mathbf{u}, p) = \mathbf{f}, \quad (11)$$

$$-\nabla \cdot \mathbf{u} = 0, \quad (12)$$

+ Initial conditions + Boundary conditions.

We try to compute the classical Benchmark: Computations of Laminar Flow Around a Cylinder form, by M. Schäfer, S. Turek, F. Durst, E. Krause, R. Rannacher

<http://www.mathematik.tu-dortmund.de/lisiii/cms/papers/SchaeferTurek1996.pdf> We compute the 2d case.

The Geometry and the physical constant are defined in file [Run:2d-data-Turek-bm.edp](#).

One of the difficulties is to obtain the correct Strouhal number of the Bénard-von Karman vortex street.

We need a high-order scheme for the time integration: we use a multi-step BDF scheme of order 1, 2 or 3: BDF1 is Euler,

BDF2 is  $\partial_t \mathbf{u} \sim \frac{3u^{n+1} - 4u^n + u^{n-1}}{2\delta t}$  and BDF3 is  $\partial_t \mathbf{u} \sim \frac{11u^{n+1} - 18u^n + 9u^{n-1} - 2u^{n-2}}{6\delta t}$

([https://en.wikipedia.org/wiki/Backward\\_differentiation\\_formula](https://en.wikipedia.org/wiki/Backward_differentiation_formula))

# Incompressible Navier-Stokes equation: unsteady computations (program)

```
real[int,int] BDF= [ [1,-1, 0,0],  
                    [3./2.,-4/2., 1/2.,0],  
                    [11./6.,-18./6., 9./6., -2./6.]];
```

to empty the file

```
{ofstream ff(datafile); }// empty file ..
```

to write in a file,

```
drag = -int1d(Th,3) ( 2*nu* ([1.,0]'*D(un1,un2))*[N.x,N.y]) - p*N.x ;  
lift  = -int1d(Th,3) ( 2*nu* ([0.,1.]'*D(un1,un2))*[N.x,N.y]) - p*N.y ;  
TCd[itime]=Cd = ccdrag*drag;  
TCl[itime]=Cl = ccdrag*lift;  
real deltap = p(xa,ya)-p(xe,ye);  
  
cout << "_Time_"<< time+dt << "_at_" << time/ccfreq << "_Cd_" << Cd << "_Cl_"  
      << Cl  
      << "_Delta_P=" << deltap << "/_max:__" << Cdx << "_" << Clx << "_" << Cpx  
      << endl;  
ofstream ff(datafile,append);  
ff << time << "_" << time/ccfreq << "_" << Cd << "_" << Cl << "_"<< deltap  
   << Cdx << "_" << Clx << "_" << Cpx <<endl;
```

Run:NS-Newton-Turek-bm.edp

# Unsteady incompressible Navier-Stokes equation: method of characteristics

For a flow field  $\mathbf{u}$  the total (or material) derivative is

$$\frac{D\mathbf{u}}{Dt} = \frac{\partial\mathbf{u}}{\partial t} + (\mathbf{u}\cdot\nabla)\mathbf{u},$$

A correct numerical scheme used to approximate  $\frac{D\mathbf{u}}{Dt}$  has to take into account the movement of a particle: let us denote by  $x^n$  (resp.  $x^{n+1}$ ) the particle position at time  $t^n$  (resp.  $t^{n+1}$ ); we can write

$$\frac{D\mathbf{u}}{Dt}(x^{n+1}) \sim \frac{\mathbf{u}^{n+1}(x^{n+1}) - \mathbf{u}^n(x^n)}{\delta t}$$

Defining the characteristic flow (passing at time  $t$  through the point  $\mathbf{x}$ )

$$\begin{cases} \frac{\partial\mathbf{X}}{\partial\tau}(\tau, t, \mathbf{x}) = \mathbf{u}(\tau, \mathbf{X}(\tau, t, \mathbf{x})), & \tau \in (0, t_{max}) \\ \mathbf{X}(t, t, \mathbf{x}) = \mathbf{x}, \end{cases} \quad (13)$$

one can express the total derivative of any function  $\Phi(t, \mathbf{x})$  as

$$\frac{D\Phi}{Dt}(t, \mathbf{x}) = \left( \frac{\partial\Phi}{\partial t} + \mathbf{u}\cdot\nabla\Phi \right) (t, \mathbf{x}) = \frac{\partial}{\partial\tau} (\Phi(\tau, \mathbf{X}(\tau, t, \mathbf{x}))) |_{\tau=t} \quad (14)$$

# Method of characteristics in FreeFem++

We use the time discretization:

$$\left(\frac{D\Phi}{Dt}\right)^{n+1}(\mathbf{x}) \approx \frac{\Phi^{n+1}(\mathbf{x}) - \Phi^n \circ \mathbf{X}^n(\mathbf{x})}{\delta t}, \quad (15)$$

with  $\mathbf{X}^n(\mathbf{x})$  a suitable approximation of  $\mathbf{X}(t_n, t_{n+1}, \mathbf{x})$ , obtained by an integration back in time of (13) from  $t_{n+1}$  to  $t_n$  for each grid point  $\mathbf{x}$ . The Galerkin characteristic method is implemented in Freefem++ as an operator computing  $\Phi \circ \mathbf{X}^n$  for given: mesh, convection velocity field and time step.

The FreeFem++ operator `convect([u1, u2], -dt, ..)` computes:

$$\frac{D\mathbf{u}}{Dt}(x^{n+1}) \sim \frac{\mathbf{u}^{n+1} - \mathbf{u}^n \circ \mathbf{X}^n}{\delta t}$$

Example: solve the convection equation with given velocity  $\mathbf{u}$

$$\partial_t a + (\mathbf{u} \cdot \nabla) a = 0, \quad + \text{initial condition}$$

```
for (int i=0; i< 20 ; i++) {  
  t += dt;   vo[]=v[];  
  v=convect([u1, u2], -dt, vo); // convect v by u1, u2, dt seconds, results in f  
  plot(v, fill=1, wait=0, dim=3, cmm="convection:_t="+t  
    + ",_min=" + v[].min + ",_max=" + v[].max); }
```

Run: `convect.edp` Exercise: use the characteristics method for the unsteady Navier-Stokes computation of the entrained cavity flow.

- 1 Linear Elasticity: weak formulations and programs (Lesson 5)
- 2 Non-linear problems (Lesson 6)
- 3 Mesh adaptation (Lesson 6)
- 4 Incompressible Fluid Dynamics (Lesson 7)
- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)

- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)
  - A free-boundary problem: modelling the water infiltration 1/2
  - Eigenvalue problems

# A free-boundary problem: modelling the water infiltration

We use a simple model to study water infiltration = the process by which water on the ground surface enters the soil.

Let  $\Omega$  be a trapezoidal domain, defined in FreeFem++ by:

```
real L=10,h=2.1 h1=0.35;           //Lenght, Left and Right Height
// trapeze
border a(t=0,L){x=t;y=0;};         // bas
border b(t=0,h1){x=L;y=t;};        // droite
border f(t=L,0){x=t;y=t*(h1-h)/L+h;}; // free surface
border d(t=h,0){x=0;y=t;};         // gauche
int n=10;
mesh Th=buildmesh (a(L*n)+b(h1*n)+f(sqrt(L^2+(h-h1)^2)*n)+d(h*n));
plot(Th);
```

## A free-boundary problem: modelling the water infiltration 2/2

The model is: find  $p$  and  $\Omega$  such that:

$$\left\{ \begin{array}{ll} -\Delta p = 0 & \text{in } \Omega \\ p = y & \text{on } \Gamma_b \\ \frac{\partial p}{\partial n} = 0 & \text{on } \Gamma_d \cup \Gamma_a \\ \frac{\partial p}{\partial n} = \frac{q}{K} n_x & \text{on } \Gamma_f \quad (\text{Neumann}) \\ p = y & \text{on } \Gamma_f \quad (\text{Dirichlet}) \end{array} \right.$$

where the input water flux is  $q = 0.02$ , and  $K = 0.5$ .

The velocity  $u$  of the water is given by  $u = -\nabla p$ .



# Modelling the water infiltration: the algorithm

We use the following fixed-point method: (*with bad main B.C. [Run:freeboundaryPB.edp](#)*)  
Let  $k = 0$ ,  $\Omega^k = \Omega$ . For the first step, we forget the Neumann B.C. and we solve the problem: find  $p$  in  $V = H^1(\Omega^k)$ , such as  $p = y$  on  $\Gamma_b^k$  and  $\Gamma_f^k$ , and

$$\int_{\Omega^k} \nabla p \nabla p' = 0, \quad \forall p' \in V \text{ with } p' = 0 \text{ on } \Gamma_b^k \cup \Gamma_f^k$$

With the **residual of the Neumann boundary condition**, we build a domain transformation  $\mathcal{F}(x, y) = [x, y - v(x)]$ , where  $v$  is solution to:  $v \in V$ , such than  $v = 0$  on  $\Gamma_a^k$  (bottom) and

$$\int_{\Omega^k} \nabla v \nabla v' = \int_{\Gamma_f^k} \left( \frac{\partial p}{\partial n} - \frac{q}{K} n_x \right) v', \quad \forall v' \in V \text{ with } v' = 0 \text{ sur } \Gamma_a^k$$

Remark: we can use the previous equation to evaluate

$$\int_{\Gamma^k} \frac{\partial p}{\partial n} v' = - \int_{\Omega^k} \nabla p \nabla v'$$

# Modelling the water infiltration: implementation

The new domain is:  $\Omega^{k+1} = \mathcal{F}(\Omega^k)$ .

Warning: if the displacement is too large we can have triangle overlapping.

```
Vh u, v, uu, vv;  
problem Pu(u, uu, solver=CG) = int2d(Th) ( dx(u) * dx(uu) + dy(u) * dy(uu) )  
  + on(b, f, u=y) ;  
problem Pv(v, vv, solver=CG) = int2d(Th) ( dx(v) * dx(vv) + dy(v) * dy(vv) )  
  + on(a, v=0) + int1d(Th, f) (vv * ((q/K) * N.y - (dx(u) * N.x + dy(u) * N.y))) ;  
real errv=1;  
while(errv > 1e-6) { j++;  
  Pu;      Pv;  
  plot(Th, u, v, wait=0);  
  errv = int1d(Th, f) (v * v);  
}
```

Here tricky code to take account the triangle overlapping

```
Th = movemesh(Th, [x, y-coef*v]); // calcul de la deformation  
cout << "\n\n" << j << "-----_errv_=" << errv << "\n\n";  
}
```

Run: [freeboundary.edp](http://freeboundary.edp)

- 5 Moving boundaries/ Eigenvalue problems / Parallel computing (Lesson 8)
  - A free-boundary problem: modelling the water infiltration 1/2
  - Eigenvalue problems

# Computing eigenvalues and eigenvectors: an example

Find the first  $\lambda, u_\lambda$  such as:

$$a(u_\lambda, v) = \int_{\Omega} \nabla u_\lambda \nabla v = \lambda \int_{\Omega} u_\lambda v = \lambda b(u_\lambda, v)$$

Boundary conditions are imposed using exact penalization: we set to  $1e30 = tgv$  the diagonal terms corresponding to locked degrees of freedom. Consequently, we impose Dirichlet boundary conditions only for the variational form of  $a$  and not for the variational form of  $b$ , because we compute eigenvalue of

$$\frac{1}{\lambda} v = A^{-1} B v$$

Otherwise we can get spurious mode.

FreeFem++ uses an Arpack interface:

```
int k=EigenValue (A, B, sym=true, value=ev, vector=eV) ;
```

# Computing eigenvalues and eigenvectors: the program

```
real sigma = 0; // value of the shift
varf a(u1,u2)= int2d(Th) ( dx(u1)*dx(u2) + dy(u1)*dy(u2) - sigma* u1*u2 )
                    + on(1,2,3,4,u1=0) ; // Boundary condition
varf b([u1],[u2]) = int2d(Th) ( u1*u2 ) ; // no Boundary condition
matrix A= a(Vh,Vh,solver=UMFPACK);
matrix B= b(Vh,Vh,solver=CG,eps=1e-20);

.....

for (int i=0;i<k;i++)
{ u1=eV[i];
  real gg = int2d(Th) (dx(u1)*dx(u1) + dy(u1)*dy(u1));
  real mm= int2d(Th) (u1*u1) ;
  real err = int2d(Th) (dx(u1)*dx(u1) + dy(u1)*dy(u1) - (eV[i])*u1*u1) ;
  if(abs(err) > 1e-6) nerr++;
  if(abs(eV[i]-eev[i]) > 1e-1) nerr++;
  cout << " _----_" << i << " " << eV[i] << " _==_" << eev[i] << " _err=_ " << err <<
    " _----_"<<endl;
  plot (eV[i],cmm="Eigen_ Vector_" + i + "_valeur_" + eV[i] ,wait=1,value=1,dim=3,
    fill=1);
}
```

Run:Lap3dEigenValue.edp

Run:LapEigenValue.edp

Run:free-cyl-3d.edp